

SWTPC 6800/CT-1024/AC-30 Cassette Tape

Diagnostic Programs

These two diagnostic programs have been written to generate and verify respectively cassette tapes generated on the SWTPC 6800 Computer System with the CT-1024 Terminal System and AC-30 cassette tape interface. Diagnostics for other hardware configurations may be written following a similar program step sequence as used in these diagnostics. The programs as they are listed generate and check data written in incremental segments where the recorder motor is actually turned off by the program between repeated recorded segments. Simple program "patches" change the programs to generate and check data written in continuous form as well.

SWTPC Cassette Test Tape Generator TAPWRT-1

This program in its unmodified form has been written to generate a test tape which checks the recorder and interface operation in the incremental (motor start/stop) mode of operation. The recorder used of course must have a "REMOTE" (motor control) jack which is fed by the control logic on the AC-30 cassette interface. The program when initiated sends a command to turn the read logic off if on and turn the record logic on. When the record logic on the cassette interface is initiated and the motor control switch is in the auto position as it should be, there is a delay provided by an adjustable one shot on the cassette interface before "marking" carrier is actually recorded on the tape. This allows the recorder motor to come up to speed before signals are recorded which guarantees the readability of the data stored on the tape. Since the program has no way of knowing when the recorder has come up to speed and the carrier has been turned on, the program has its own software delay one shot subroutine called DELAY which must be slightly greater in length than the delay time of the electrical one shot on the cassette interface. This delay time is set by loading the hexadecimal delay time from the user set data at address A01B. The delay time is approximately 0.2 seconds times the value stored in this address. It has been arbitrarily chosen in this program to be 0F or $(0.2 \text{ seconds} \times 0F) = (0.2 \times 16) = 3.2 \text{ seconds}$. If you ever write your own programs using the cassette interface in the incremental (start/stop) mode you will probably want to incorporate this subroutine into your own program. You must provide ample start-up delay between the time the recorder motor is turned on and the time "marking" carrier and data is recorded on the tape.

The program then sequentially writes ASCII characters 40₁₆ thru 60₁₆ inclusive: @ABCDEFGHIJKLMN₁₆OPQRSTUVWXYZC\]^_ . These characters were chosen because they are all visible to the CT-1024 terminal system when the AC-30 cassette interface is operated in local mode. The program then sends a Control T(14₁₆) which although recorded on the tape is not displayable by the CT-1024 terminal system since it does not display control characters. This character simultaneously turns off the recorded "marking" carrier and the recorder's motor. After which another adjustable delay loop located at address A02E is provided to give the recorder time to stop. Whether or not you need this second delay loop in your own programs will depend upon whether you will be doing incremental (start/stop) or continuous reading of your recorded tape. If you will be doing incremental reading use the delay, if continuous, it is unnecessary. After the delay the program then branches back to the beginning and repeats itself indefinitely until the system is powered down or

reset. Allowing the program to record data on one side of a spare cassette will give you a convenient cassette interface test tape.

If your system does not have motor control or if you do not wish to check for motor control or if you just want to load a suspicious tape to check for data dropouts you can disable the motor control feature simply by flipping the motor control switch on the AC-30 cassette interface to the manual position. To remove the now unnecessary software delays, use the MIKBUG memory change instruction to change the data in memory locations A01C, and A01D, to 01_{16} (No Operation) instructions before generating the test tape.

The program itself uses 68_{10} words and is loaded within the 128 word RAM used by the MIKBUG operating system on the MP-A Microprocessor/System Board. A program may reside in external RAM memory simultaneously with the program loaded within the 128 word RAM, or the program may be run with no MP-M memory boards installed on the system at all. The program may be loaded either from tape or instruction by instruction using MIKBUG starting from address A014 thru A033, and then from address A048 thru A06B. The program must be loaded in two parts to avoid interfering with the system's push down stack.

To Generate an Incremental (Start/Stop) Tape

1. Load the TAPWRT-1 program into computer memory as described in the preceding instructions.
2. Make sure you have MIC, EAR and MOTOR CONTROL connections between the AC-30 cassette interface and the recorder you wish to use.
3. Set the AC-30 switches so the Motor Control switch is on Manual with the Record Select switch flipped to drive the correct set of jacks. Flip the Record OFF/ON and READ OFF/ON switches to clear the Ready lights if they are ON. The switches must then be returned to their normal center position. The Local/Remote switch should be set on Remote and the Power switch must of course be ON.
4. Insert a previously erased tape into your recorder and rewind to the beginning of the tape. Put the recorder in the Play mode for about 20 seconds to advance past the leader on the tape. Stop the recorder.
5. Flip the Motor Control switch to the AUTO position.
6. Put your recorder in the Record mode. The tape should not move since the recorder motor is now controlled by the AC-30 interface and your computer.
7. When you are ready for the program to start, enter a G for go on your terminal's keyboard. The tape should move for a few seconds, stop, then move again, repeatedly until stopped. All data written to tape will be simultaneously displayed on the terminal's display.

Since the program counter is set when the program is initially loaded, the program is initiated as described in the "Go to User's Program" section of the Engineering Note 100. Once initiated, the program can be stopped only by depressing the "RESET" button. The program may then be re-started after resetting the program counter to A014 as described in the "Display contents of MPU Registers Function" section of Engineering Note 100.

MIKBUG® is a registered trademark of Motorola Inc.

To Generate a Continuous Data Tape

1. Load the TAPWRT-1 program into computer memory as described in the preceding instructions.
2. Using the MIKBUG memory change function, change the data in memory locations A01C, and A01D to 01_{16} (No Operation) instructions.
3. Make sure you have MIC, EAR and if you wish, MOTOR CONTROL connections between the AC-30 cassette interface and the recorder you will be using.
4. Set the AC-30 switches so the Motor Control switch is on Manual with the Record Select switch flipped to drive the correct set of jacks. Flip the Record OFF/ON and READ OFF/ON switches to clear the Ready lights if they are ON. The switches must then be returned to their normal center position. The Local/Remote switch should be set on Remote and the Power switch must of course be ON.
5. Insert a previously erased tape into your recorder and rewind to the beginning of the tape. Put the recorder in the Play mode for about 20 seconds to advance past the leader on the tape. Stop the recorder.
6. Put your recorder in the Record mode. The tape should move since the recorder motor is under manual control.
7. After recording about 10 seconds of "marking" data, enter a G for go on your terminal's keyboard. The tape should move continuously to the end of the tape unless stopped sooner. All data written to tape will be simultaneously displayed on the terminal's display.

Since the program counter is set when the program is initially loaded, the diagnostic is initiated as described in the "Go to User's Program" section of the Engineering Note 100. Once initiated, the program can be stopped only by depressing the "RESET" button. The program may then be re-started after resetting the program counter to A014 as described in the "Display contents of MPU Registers Function" section of Engineering Note 100.

NAM TAPWRT-1
 *SWTPC Cassette Test Tape Generator

```

E1D1            OUTEEE EQU     $E1D1
E07E            PDATA1 EQU    $E07E

A014                            ORG     $A014
A014 CE A0 69    START    LDX    #OFF
A017 BD E0 7E                    JSR    PDATA1
A01A 86 0F                        LDA A    #$0F
A01C 8D 2F                        BSR    DELAY
A01E C6 40                        LDA B    #$40
A020 17                    LOOP2    TBA
A021 BD E1 D1                    JSR    OUTEEE
A024 5C                            INC B
A025 C1 60                        CMP B    #$60
A027 26 F7                        BNE    LOOP2
A029 86 14                        LDA A    #$14            (record off)
A02B BD E1 D1                    JSR    OUTEEE
A02E 86 08                        LDA A    #$08
A030 8D 1B                        BSR    DELAY
A032 20 E0                        BRA    START

A048                            ORG     $A048
A048 A0 14                        FDB    START
A04A                        TMP    RMB    3

A04D CE A0 4A    DELAY    LDX    #TMP
A050 6F 00                        CLR    0,X
A052 6F 01                        CLR    1,X
A054 6F 02                        CLR    2,X
A056 6C 00                    LOOP3    INC    0,X
A058 2A FC                        BPL    LOOP3
A05A 6F 00                        CLR    0,X
A05C 6C 01                        INC    1,X
A05E 2A F6                        BPL    LOOP3
A060 6F 01                        CLR    1,X
A062 6C 02                        INC    2,X
A064 A1 02                        CMP A    2,X
A066 26 EE                        BNE    LOOP3
A068 39                            RTS

A069 13                    OFF    FCB    $13,$12,$04
A06A 12 04
  
```

END

NO ERROR(S) DETECTED

SYMBOL TABLE:

```

DELAY  A04D    LOOP2  A020    LOOP3  A056    OFF    A069    OUTEEE  E1D1
PDATA1 E07E    START  A014    TMP    A04A
  
```

The program listings in the original document were hand assembled with a few syntax errors. These listing are from the TSC 6800 Assembler. The resulting opcodes are identical with the original.

SWTPC Cassette Tape Read Diagnostic TAPRED-1

This program reads and verifies tapes generated by the TAPWRT-1 program. The tapes may be either incremental or continuous. This program checks both types without program modifications. The first thing the program does is disable the computer's control interface echo. This prevents the tape read data from being displayed on the screen during the program. It then turns the cassette interface's read circuitry ON and looks for an @ to be read from the tape. When it reads one, it looks for all subsequent characters to be read in correct order as written by the TAPWRT-1 program. If they are, it outputs a / to the terminal's display for each string of characters read. If any one of the characters is read incorrectly including the control T (14₁₆) which follows each string of alphanumeric data, an X is output to the terminal's display. The program repeats this verification sequence until the tape is exhausted or the computer system is reset. If an error is found, the tape can be manually backed up several frames and the data displayed on the terminal's display with the AC-30 cassette interface operated in the Local mode for manual verification. Consistent error data on repeated reads indicates a record error while inconsistent error data on repeated reads indicates a read error.

This program should initially be used to verify proper operation of the AC-30 cassette interface but may be used to check the reliability of the tapes themselves once the reliability of the interface has been established. When checking for bad tapes, the test data should be recorded continuously not incrementally for thorough testing.

The program itself uses 88₁₀ words and is loaded within the 128 word RAM used by the MIKBUG operating system on the MP-A Microprocessor /System Board. A program may reside in external RAM memory simultaneously with the diagnostic loaded within the 128 word RAM, or the diagnostic may be run with no MP-M memory boards installed on the system at all. The diagnostic may be loaded either from tape or instruction by instruction using MIKBUG starting from address A048, thru A07E.

To Read a Test Tape Generated by The TAPWRT-1 Program

1. Load the TAPRED-1 program into computer memory as described in the preceding instructions.
2. Make sure you have MIC, EAR and if you wish MOTOR CONTROL connections between the AC-30 cassette interface and the recorder you wish to use.
3. Set the AC-30 switches so the Motor Control switch is on Manual with the RECORD SELECT switch flipped to drive the correct set of jacks. Flip the RECORD OFF/ON and READ OFF/ON switches to clear the Ready lights if they are ON. The switches must then be returned to their normal center position. The Local/Remote switch should be set on Remote and the Power switch must of course be ON.
4. Insert a previously recorded TAPWRT-1 tape into your recorder and rewind to the beginning of the tape. Stop the recorder.
5. When you are ready for the program to start, enter a G for go on your terminal's keyboard and put your recorder in the play mode. The tape should move and after the first block of data is read either a / or X will be displayed on the terminal's display. The / indicates a correct read while an X indicates an incorrect one.

Since the program counter is set when the program is initially loaded, the diagnostic is initiated as described in the "Go to User's Program" section of the Engineering Note 100. Once initiated, the program can be stopped only by pressing the "RESET" button. The program may then be re-started after resetting the program counter to A04A as described in the "Display contents of MPU Registers Function" section of Engineering Note 100.

MIKBUG is a registered trademark of Motorola Inc.

```

NAM      TAPRED-1
* SWTPC  Cassette Tape Read Diagnostic
E1D1    OUTEEE EQU  $E1D1
E1AC    INEEEE EQU  $E1AC

A014    ORG      $A014
A014    TMP      RMB      3
A017 C6 02    DELOUT LDA B  #$02
A019 CE A0 14    LDX      #TMP
A01C 6F 00      CLR      0,X
A01E 6F 01      CLR      1,X
A020 6F 02      CLR      2,X
A022 6C 00    LOOPA INC      0,X
A024 2A FC      BPL      LOOPA
A026 6F 00      CLR      0,X
A028 6C 01      INC      1,X
A02A 2A F6      BPL      LOOPA
A02C 6F 01      CLR      1,X
A02E 6C 02      INC      2,X
A030 E1 02      CMP B   2,X
A032 26 EE      BNE      LOOPA
A034 BD E1 D1    JSR      OUTEEEE
A037 39      RTS

A048    ORG      $A048
A048 A0 4A      FDB      START
A04A 86 3C    START LDA A  #$3C
A04C B7 80 07    STA A  $8007
A04F 86 11      LDA A  #$11      (read on)
A051 BD E1 D1    JSR      OUTEEEE
A054 C6 40    REDO  LDA B  #'@
A056 BD E1 AC  LOOP1 JSR      INEEEE
A059 11      CBA
A05A 26 FA      BNE      LOOP1
A05C 5C    LOOP2 INC B
A05D C1 60      CMP B  #$60
A05F 27 08      BEQ      RECOFF
A061 BD E1 AC  CYCLE JSR      INEEEE
A064 11      CBA
A065 26 12      BNE      ERROR
A067 20 F3      BRA      LOOP2
A069 C6 14    RECOFF LDA B  #$14
A06B BD E1 AC  JSR      INEEEE
A06E 11      CBA
A06F 26 08      BNE      ERROR
A071 86 2F      LDA A  #' /
A073 8D A2      BSR      DELOUT
A075 C6 40      LDA B  #'@
A077 20 E8      BRA      CYCLE
A079 86 58    ERROR LDA A  #'X
A07B 8D 9A      BSR      DELOUT
A07D 20 D5      BRA      REDO
END

```

NO ERROR(S) DETECTED

SYMBOL TABLE:

CYCLE	A061	DELOUT	A017	ERROR	A079	INEEEE	E1AC	LOOP1	A056
LOOP2	A05C	LOOPA	A022	OUTEEEE	E1D1	RECOFF	A069	REDO	A054
START	A04A	TMP	A014						

SWTPC 6800/AC-30 Calibrate Tape Generator Program FIVPNT-1 and Calibration Procedure

To set the calibration adjustment on the AC-30 cassette interface it is wise to generate a calibration tape containing a continuous string of ASCII fives. An ASCII five (35₁₆) has a bit sequence of 0011 0101 which when combined with the UART's start and stop bits gives a good alternating bit pattern for calibrating the cassette interface. The program presented here also records a non-displayable rubout (7F₁₆) between fives to aid in resynchronization of data should an incorrect read ever take place. An alternative to this program is to operate the cassette interface in the LOCAL mode and manually thru the terminal's keyboard enter the fives. There is no need to enter the rubouts when the fives are written to tape this way. To generate a calibration tape on a SWTPC 6800 Computer/CT-1024 Terminal System with the cassette interface connected between the two use the following procedure:

1. Load the FIVPNT-1 program into computer memory.
2. Make sure you have MIC, EAR and if you wish, MOTOR CONTROL connections between the AC-30 cassette interface and the recorder you will be using.
3. Set the AC-30 switches so the Motor Control switch is on Manual with the Record Select switch flipped to drive the correct set of jacks. Flip the Record OFF/ON and READ OFF/ON switches to clear the Ready lights if they are ON. The switches must then be returned to their normal center position. The Local/Remote switch should be set on Remote and the Power switch must of course be ON.
4. Insert a previously erased tape into your recorder and rewind to the beginning of the tape. Put the recorder in the Play mode for about 20 seconds to advance past the leader on the tape. Stop the recorder.
5. Put your recorder in the Record mode. The tape should move since the recorder motor is under manual control.
6. After recording about 10 seconds of "marking" data, enter a G for go on your terminal's keyboard. The tape should move continuously to the end of the tape unless stopped sooner. All data written to tape will be simultaneously displayed on the terminal's display. If your system's record status light does not come on after entering the G then the terminal system did not decode the received RECORD ON command and forward it to the cassette interface. If you system is wired to decode these commands then something is wrong, if it is not wired for decode then you will need to manually set the RECORD latch using the front panel toggle switch.

Since the program counter is set when the program is initially loaded, the diagnostic is initiated as described in the "Go to User's Program" section of the Engineering Note 100. Once initiated, the program can be stopped only by depressing the "RESET" button. The program may then be re-started after resetting the program counter to A04F as described in the "Display contents of MPU Registers Function" section of Engineering Note 100.

NAM FIVPNT-1
* SWTPC 6800 Calibrate Tape Generator Program

```
E07E          PDATA1 EQU    $E07E

A048          ORG    $A048
A048 A0 4F          FDB    START
A04A 13          STR5    FCB    $13,$12,$7F,$35,$04
A04B 12 7F
A04D 35 04

A04F CE A0 4A  START  LDX    #STR5
A052 BD E0 7E  LOOP1  JSR    PDATA1
A055 09          DEX
A056 09          DEX
A057 20 F9          BRA    LOOP1

END
```

NO ERROR(S) DETECTED

SYMBOL TABLE:

LOOP1 A052 PDATA1 E07E START A04F STR5 A04A

To calibrate your cassette interface, you will need to read back the just generated calibration tape from your terminal:

1. Set the AC-30 switches so the Motor Control switch is on Manual with the Read select switch flipped to feed the correct set of jacks. Flip the Record OFF/ON and Read OFF/ON switches so as to clear the Record ready and set the Read ready latches. The Local/Remote switch should be set on Local and the power switch must of course be ON.
2. Be sure your terminal's framing (stop-bit) error detection circuits are not disabled for this test to insure accurate calibration.
3. Insert the previously recorded calibrate tape into the selected Read recorder and rewind to the beginning of the tape. Put the recorder in the Play mode and watch the terminal's screen. When data starts appearing on the screen adjust trimmer resistor R16 midway between errored read settings of the control. Setting the trimmer too far clockwise will cause errors just as will setting it too far counterclockwise. Proper operation is indicated by reading a continuous string of 5's.

If you are using the SWTPC 6800 Computer system with something other than an RS-232 300 baud terminal and the cassette interface is not installed in series between the terminal and computer, then the diagnostic will have to be rewritten to operate thru a separate selected interface. Reading the cassette tape back to the terminal in the LOCAL mode probably will not be applicable either so you will need to write a computer program to read the data from the tape and in some way tell you if it is being read accurately. Be sure to check incoming characters for the proper reception of the stop bit when you are reading tape data. Trimmer resistor R16 is adjusted just as before.

If you are using another computer system with the SWTPC CT-1024 terminal system or equivalent RS-232 300 baud terminal then all you will need to do is re-write the FIVPNT-1 program using your computer's own instruction set.